Laplace’s Equation in a Disk

The statement of the problem (Subsection 2.5.2 in the book):

In this file I will consider the Laplace's equation in a disk. See Subsection 2.5.2 (page 73) in the book. The equation is

$$\nabla^2 u = \frac{1}{r}\frac{\partial}{\partial r}\left(r\,\frac{\partial u}{\partial r}\right) + \frac{1}{r^2}\frac{\partial^2 u}{\partial \theta^2} = 0, \quad 0 < r \leq R, \quad -\pi \leq \theta \leq \pi,$$

$$\theta \text{ "boundary conditions" } \begin{cases} u(r, -\pi) - u(r, \pi) = 0, \\ \frac{\partial u}{\partial \theta}(r, -\pi) - \frac{\partial u}{\partial \theta}u(r, \pi) = 0, \end{cases}$$

$$r \text{ boundary conditions } \begin{cases} u(R, \theta) = f(\theta), \quad 0 \leq \theta \leq 2\pi, \\ |\,u(0, \theta)\,| < \infty. \end{cases}$$

We will solve this boundary value problem by the separation of variables method. We look for the solution of the form $u(r, \theta) = A(r)\,B(\theta)$. This leads to :

$$\nabla^2 u = \left(\frac{1}{r}\frac{d}{dr}\left(r\,\frac{d}{dr}A(r)\right)\right)B(\theta) + \frac{1}{r^2}A(r)\,\frac{d^2}{d\theta^2}B(\theta) = 0,$$

$$B(-\pi) - B(\pi) = 0,$$

$$B'(-\pi) - B'(\pi) = 0.$$

Here, as before, we ignore the nonhomogeneous set of boundary conditions. Separating variables we obtain (We divide by $\frac{1}{r^2}A(r)\,B(\theta)$)

$$\frac{\frac{1}{r}\frac{d}{dr}\left(r\,\frac{d}{dr}A(r)\right)}{\frac{1}{r^2}A(r)} = -\frac{\frac{d^2}{d\theta^2}B(\theta)}{B(\theta)} = \lambda,$$

what leads to the boundary eigenvalue problem for the function $B(\theta)$:

$$-\frac{d^2}{d\theta^2}B(\theta) = \lambda\,B(\theta),$$

$$B(-\pi) - B(\pi) = 0,$$

$$B'(-\pi) - B'(\pi) = 0,$$

and the equation

$$r \frac{d}{dr}\left(r \frac{d}{dr} A(r)\right) - \lambda A(r) = 0,$$

for the function *A*.

The boundary eigenvalue problem for the function *B* is identical to the problem studied in 2.5.1 (thin circular ring). The eigenvalues of this problem and the corresponding eigenfunctions are given by:

The eigenvalues: The corresponding eigenfunction(s):

$\lambda_0 = 0$, 1 (the constant function),
$\lambda_1 = 1^2$, Sin[θ], and Cos[θ] (two linearly independent eigenfunctions),
$\lambda_2 = 2^2$, Sin[2 θ], and Cos[2 θ] (two linearly independent eigenfunctions),

and in general:

$\lambda_n = n^2$, Sin[$n\,\theta$], and Cos[$n\,\theta$], $n \in \mathbb{N} \cup \{0\}$.

Note that the last formula makes sense for $n = 0$, since Cos[0 x] = 1, and Sin[0 x] = 0 (which can be ignored).

Solving the equation for the function *A*:

We conclude that the solutions for the boundary eigenvalue problem for the function *B* are numbers $\lambda = n^2$ (eigenvalues) and the corresponding functions Cos[$n\,x$], Sin[$n\,x$], $n \in \mathbb{N} \cup \{0\}$ (eigenfunctions).
For these values $\lambda = n^2$ we have to solve the equation for the function *A*:

$$r \frac{d}{dr}\left(r \frac{d}{dr} A(r)\right) - n^2 A(r) = 0,$$

or in a more transparent form

$$r^2 \frac{d^2}{dr^2} A + r \frac{d}{dr} A - n^2 A = 0.$$

Let us show how to use *Mathematica* to solve this equation. The basic command is

In[9]:=
```
DSolve[r^2 A''[r] + r A'[r] - n^2 A[r] == 0, A[r], r]
```

Out[9]= $\{\{A[r] \to c_1 \text{Cosh}[n \text{Log}[r]] + i\, c_2 \text{Sinh}[n \text{Log}[r]]\}\}$

But this is not a desirable form of the solution; there must be a simpler expression for

In[10]:=
```
TrigToExp[{Cosh[n Log[r]], Sinh[n Log[r]]}]
```

Out[10]= $\left\{\frac{r^{-n}}{2} + \frac{r^n}{2}, -\frac{r^{-n}}{2} + \frac{r^n}{2}\right\}$

Since any linear combinations of solutions is a solution we have that

In[11]:= $\{\{1, 1\}, \{1, -1\}\}.\left\{\frac{r^{-n}}{2} + \frac{r^n}{2}, -\frac{r^{-n}}{2} + \frac{r^n}{2}\right\}$

Out[11]= $\{r^n, r^{-n}\}$

are also solutions. Thus, *Mathematica* should have given the general solution

In[12]:= $C[1]\, r^n + C[2]\, r^{-n}$

Out[12]= $r^n c_1 + r^{-n} c_2$

Below, just to demonstrate some *Mathematica* commands, I will try to force *Mathematica* to give this solution.

In[13]:=
```
Collect[Expand[Simplify[TrigToExp[
    DSolve[r^2 A''[r] + r A'[r] - n^2 A[r] == 0, A[r], r][[1]][[1]][[2]]], r > 0]], r^n]
```

Out[13]= $r^{-n}\left(\frac{c_1}{2} - \frac{i\, c_2}{2}\right) + r^n\left(\frac{c_1}{2} + \frac{i\, c_2}{2}\right)$

In the last expression constants are complex numbers and we can replace them with arbitrary constants *C*[1] and *C*[2] and we can write this general solution as:

In[14]:=
```
solA =
 Collect[Expand[Simplify[TrigToExp[
      DSolve[r^2 A''[r] + r A'[r] - n^2 A[r] == 0, A[r], r][[1]][[1]][[2]]], r > 0]],
   r^n] /. {(-1/2 I C[2] + C[1]/2) -> C[1], (1/2 I C[2] + C[1]/2) -> C[2]}
```

Out[14]= $r^{-n} c_1 + r^n c_2$

Note that the above solution is the general solution of the equation for A only if $n > 0$. Namely, for the above formula to be the general solution the functions r^n and r^{-n} must be linearly independent on $0 \leqslant r \leqslant R$. For $n = 0$ this obviously is not a case. Thus we have to solve the equation for $n = 0$ separately:

In[15]:=
```
DSolve[r^2 A''[r] + r A'[r] == 0, A[r], r][[1]][[1]][[2]]
```

Out[15]= $c_2 + c_1 \text{Log}[r]$

Nice solution! Now we recall the boundary condition for $u(r, \theta)$ near $r = 0$:

$| u(0, \theta) | < \infty$, or in terms of A : $| A(\text{near } 0) | < \infty$.

This clearly eliminates the functions r^{-n} and Log[r] as possible solutions for $A(r)$. Thus the only possible solutions for A are

$A_n(r) = r^n, \; n \in \mathbb{N} \cup \{0\}$.

Conclusion (the solution of Laplace's equation):

We have found infinitely many special solutions for the Laplace's equation on a circular disk:

$1, \; \frac{r^n}{R^n} \text{Sin}[n\,\theta], \; \frac{r^n}{R^n} \text{Cos}[n\,\theta]$ where $n \in \mathbb{N}$.

Using the principle of superposition we conclude that the function

$$u(r, \theta) = a_0 + \sum_{n=1}^{\infty} a_n \frac{r^n}{R^n} \text{Cos}[n\,\theta] + \sum_{n=1}^{\infty} b_n \frac{r^n}{R^n} \text{Sin}[n\,\theta]$$

is also a solution. We just need to determine the constants a_n and b_n in such a way that the boundary conditions

$u(R, \theta) = f(\theta), \; 0 \le \theta \le 2\pi$

is satisfied. Put $r = R$ in the above formula for $u(r, \theta)$ and we get

$$f(\theta) = u(R, \theta) = a_0 + \sum_{n=1}^{\infty} a_n \text{Cos}[n\,\theta] + \sum_{n=1}^{\infty} b_n \text{Sin}[n\,\theta]$$

.

The orthogonality of the functions Sin[$n\,x$] and Cos[$n\,x$] leads to the formula for a_n:

$$\int_{-\pi}^{\pi} f(\theta) \,\text{Cos}[n\,\theta] \,dx \quad . \\ = a_n \int_{-\pi}^{\pi} \text{Cos}[n\,\theta] * \text{Cos}[n\,\theta] \,d\theta$$

Thus:

$$a_n = \frac{\int_{-\pi}^{\pi} f(\theta) \,\text{Cos}[n\,\theta] \,d\theta}{\int_{-\pi}^{\pi} \text{Cos}[n\,\theta] * \text{Cos}[n\,\theta] \,d\theta}, \; n \in \mathbb{N} \cup \{0\} .$$

Similarly:

$$b_n = \frac{\int_{-\pi}^{\pi} f(\theta)\,\text{Sin}[n\,\theta]\,d\theta}{\int_{-\pi}^{\pi} \text{Sin}[n\,\theta] * \text{Sin}[n\,\theta]\,d\theta},\; n \in \mathbb{N}.$$

Since we can calculate:

In[16]:=
```
FullSimplify[{∫_{-π}^{π} Cos[0 x] * Cos[0 x] ⅆx, ∫_{-π}^{π} Cos[n x] * Cos[n x] ⅆx,
  ∫_{-π}^{π} Sin[n x] * Sin[n x] ⅆx}, And[n ∈ Integers, n > 0]]
```

Out[16]= $\{2\pi, \pi, \pi\}$

we can rewrite our formulas for a_n and b_n as

$a_0 = \frac{1}{2\pi} \int_{-\pi}^{\pi} f(\theta)\,d\theta$ and

$a_n = \frac{1}{\pi} \int_{-\pi}^{\pi} f(\theta)\,\text{Cos}[n\,\theta]\,d\theta,\; n \in \mathbb{N}$ and

$b_n = \frac{1}{\pi} \int_{-\pi}^{\pi} f(\theta)\,\text{Sin}[n\,x]\,d\theta,\; n \in \mathbb{N}.$

Thus the solution of the heat equation for a circular disk is:

$$u(r,\,\theta) = a_0 + \sum_{n=1}^{\infty} a_n \frac{r^n}{R^n}\,\text{Cos}[n\,\theta] + \sum_{n=1}^{\infty} b_n \frac{r^n}{R^n}\,\text{Sin}[n\,\theta]$$

with a_0, a_n and b_n given by the above formulas.

Next we implement these formulas in *Mathematica:*

Mathematica implementation of the solution: Example 1

Here is our function $f(\theta)$ which we will approximate with the first 15 (or nn) terms in its Fourier Series.

In[17]:=
```
Clear[ff];
ff[θ_] = (θ^2 - π^2)^2 (θ + π - 2) Exp[-θ - π]
```

Out[18]= $e^{-\pi-\theta}\,(-2+\pi+\theta)\,(-\pi^2+\theta^2)^2$

Test that the function is continuous on the unit circle and that it has continuous derivative.

In[19]:=
```
test = {(ff[θ] /. {θ → -π}), (ff[θ] /. {θ → π}),
   (D[ff[θ], θ] /. {θ → -π}), (D[ff[θ], θ] /. {θ → π})}
```

Out[19]=
```
{0, 0, 0, 0}
```

In[20]:=
```
Plot[{ff[θ]}, {θ, -π, π}, Ticks → {Range[-Pi, Pi, Pi / 4], Automatic}]
```

Out[20]=



Since the function that we have chosen is quite complicated we will calculate the Fourier coefficients numerically.

In[21]:=
```
Clear[rR, nn, las, lbs];

rR = 1;
nn = 15;

las = Table[1/π * NIntegrate[Expand[ff[θ] * Cos[n θ]], {θ, -π, π}, MaxRecursion → 20,
PrecisionGoal → 12, WorkingPrecision → 20, AccuracyGoal → 12],
        {n, 1, nn}];
lbs =
   Table[1/π * NIntegrate[Expand[ff[θ] * Sin[n θ]], {θ, -π, π}, MaxRecursion → 20,
PrecisionGoal → 12, WorkingPrecision → 20, AccuracyGoal → 12],
        {n, 1, nn}];
```

In[26]:=
```
las
```

Out[26]=
```
{4.2098297814352441936, 1.8850735107510225977, -1.3177319751745400631,
 0.63118576595389593016, -0.31285535166824405226, 0.16733290545008752458,
 -0.096141798864534659058, 0.058688317311540081257, -0.037671857624295484892,
 0.025213140485440768263, -0.017476325562215542092, 0.012478426799177674280,
 -0.0091389697956058356685, 0.0068416805598380954533, -0.0052207612160820613339}
```

In[27]:=
```
lbs
```

Out[27]=
```
{3.4424524980389388569, -2.4548128011253300031, 0.22396309317629304507,
 0.15650320412268986517, -0.17075621680756197890, 0.13366272417001792630,
 -0.098980928159225305403, 0.073211786123435406775, -0.054898910066279184274,
 0.041894936228131709603, -0.032542249366315875581, 0.025701229347323929000,
 -0.020608862265404008704, 0.016753638710492858554, -0.013788685804928454428}
```

We did not include the coefficient with the constant 1. So we do it in the final formula for the solution function which I call uu.

In[28]:=
```
Clear[uu];

uu[r_, θ_] = 1/(2 π) NIntegrate[ff[θ], {θ, -π, π}, MaxRecursion → 20,
PrecisionGoal → 12, WorkingPrecision → 20, AccuracyGoal → 12] +
    Sum[las[[n]] * r^n/rR^n * Cos[n θ], {n, 1, nn}] +
    Sum[lbs[[n]] * r^n/rR^n * Sin[n θ], {n, 1, nn}];
```

In[30]:=
```
uu[.5, π / 2]
```

Out[30]= 0.857122

In[31]:=
```
uu[0, 0]
```

Out[31]= -0.39722626185151339514

Test of the approximation:

In[32]:=
```
Plot[{ff[θ], uu[rR, θ]}, {θ, -π, π},
 PlotStyle → {{Thickness[0.007], Blue}, {Thickness[0.004], Red}}]
```

Out[32]=



Quite good, visually! But let us check the absolute value of the difference

In[33]:=
```
Plot[Abs[ff[θ] - uu[rR, θ]], {θ, -π, π},
 PlotStyle → {{Thickness[0.004], Red}}, PlotRange → All]
```

Out[33]=



I find maximum and minimum of ff to set the range of plots correctly.

In[34]:=
```
FindMaximum[ff[θ], {θ, -1}]
```

Out[34]=
```
{4.83582, {θ → -0.0995197}}
```

In[35]:=
```
FindMinimum[ff[θ], {θ, -3}]
```

Out[35]=
```
{-10.5029, {θ → -2.25877}}
```

I experimented and found out that the following view point works well for the plot below.

In[36]:=
```
vp = {-1.9533572861214523`, -2.405091682993332`, 1.3602486203457567`}
```

Out[36]=
```
{-1.95336, -2.40509, 1.36025}
```

In[37]:=
```
pluu = ParametricPlot3D[ {r Cos[θ], r Sin[θ], uu[r, θ]},
        {r, 0, rR}, {θ, -π, π}, Mesh → False,
    PlotRange → {{-1, 1}, {-1, 1}, {-11, 5}}, BoxRatios → {1, 1, 1},
    PlotPoints → {20, 50}, ImageSize → 300, ViewPoint → Dynamic[vp]]
```

Out[37]=



I want to show the unit circle on the graph as well. So, I define it separately. Below is the unit circle placed at the level below the minimum temperature.

In[38]:=
```
uc = ParametricPlot3D[{rR Cos[θ], rR Sin[θ], -10.9},
   {θ, -Pi, 2 Pi}, PlotStyle → Thickness[0.005]]
```

Out[38]=



In[39]:=
```
ic = ParametricPlot3D[{rR Cos[θ], rR Sin[θ], ff[θ]},
   {θ, -Pi, Pi}, PlotStyle → {Thickness[0.01], RGBColor[0, 1, 0]},
   PlotPoints → 50, BoxRatios → {1, 1, 1}, ViewPoint → vp]
```

Out[39]=



In[40]:=
```
Show[pluu, ic, uc, BoxRatios → {1, 1, 1}, ViewPoint → vp]
```

Out[40]=



I do not know how to do a density plot in polar coordinates in *Mathematica.* So, I fool *Mathematica* to think that it works in rectangular coordinates. Interestingly, *Mathematica* has a function ArcTan as a function of two variables which will give exactly the angle θ. The only problem is that this function is not defined at {0,0}, so I define it to be 0 at the origin.

In[41]:=
```
arctan[0., 0.] = 0;
arctan[x_, y_] := ArcTan[x, y];
DensityPlot[ uu[ Sqrt[x^2 + y^2] , arctan[x, y]],
      {x, -1, 1}, {y, -1, 1}, PlotRange → {-10.6, 4.9},
 PlotPoints → {20, 20}, RegionFunction → Function[{x, y, z}, Norm[{x, y}] < 1.],
 Axes → False, Frame → True, LightingAngle →  {0, Pi / 6}, ImageSize → 300]
```

Out[41]=



This is the density plot with the value of the function similar to what I used on the website for the diffusion of dye.

In[42]:=
```
DensityPlot[uu[Sqrt[x^2 + y^2], arctan[x, y]],
    {x, -1, 1}, {y, -1, 1}, PlotRange → {-10.6, 4.9}, PlotPoints → {30, 30},
  RegionFunction → Function[{x, y, z}, Norm[{x, y}] < 1.], Axes → False,
  Frame → True, ColorFunction → (RGBColor[1, 1 - #^2, 1 - #^2] &),
  LightingAngle → {0, Pi / 3}, Epilog → {Circle[]}, ImageSize → 300]
```

Out[42]=



Mathematica implementation of the solution: Example 2

Here is our function $f(\theta)$ which we will approximate with the first 15 (or nn) terms in its Fourier Series.

In[43]:=
```
Clear[ff2];
ff2[θ_] = Abs[θ]
```

Out[44]= Abs[θ]

Test that the function is continuous on the unit circle and that it has continuous derivative.

In[45]:=
```
test = FullSimplify[{(ff2[θ] /. {θ → -π}), (ff2[θ] /. {θ → π}),
   (D[ff2[θ], θ] /. {θ → -π}), (D[ff2[θ], θ] /. {θ → π})}]
```

Out[45]= {π, π, -1, 1}

So, the derivative is not continuous, but that should not be a problem

In[46]:=
```
Plot[{ff2[θ]}, {θ, -π, π}, Ticks → {Range[-Pi, Pi, Pi / 4], Range[-Pi, Pi, Pi / 4]}]
```

Out[46]=



Since the function that we have chosen is simple we will calculate the Fourier coefficients exactlylly.

In[47]:=
```
FullSimplify[1/π * Integrate[Expand[ff2[θ] * Cos[n θ]], {θ, -π, π}], And[n ∈ Integers]]
```

Out[47]= $\frac{2\left(-1+(-1)^{n}\right)}{n^{2}\pi}$

As we can see the coefficients for even n are equal to 0, while the coefficients for odd n are -4/(n^2 Pi).

In[48]:=
```
Table[2 (-1 + (-1)^n)/(n^2 π), {n, 1, 20}]
```

Out[48]= $\left\{-\frac{4}{\pi}, 0, -\frac{4}{9\pi}, 0, -\frac{4}{25\pi}, 0, -\frac{4}{49\pi}, 0, -\frac{4}{81\pi}, 0, -\frac{4}{121\pi}, 0, -\frac{4}{169\pi}, 0, -\frac{4}{225\pi}, 0, -\frac{4}{289\pi}, 0, -\frac{4}{361\pi}, 0\right\}$

In[49]:=
```
FullSimplify[1/π * Integrate[Expand[ff2[θ] * Sin[n θ]], {θ, -π, π}], And[n ∈ Integers]]
```

Out[49]= 0

This is the approximation for the solution uu2 with 20 terms.

In[50]:=
```
rR2 = 1; nn2 = 20; Clear[uu2];

uu2[r_, θ_] = 1/(2 π) Integrate[ff2[θ], {θ, -π, π}] +
    Sum[-4/((2 k - 1)^2 Pi) * r^(2 k - 1)/rR2^(2 k - 1) * Cos[(2 k - 1) θ], {k, 1, nn2}];
```

In[52]:=
```
uu2[.5, π / 2]
```

Out[52]= 1.5708

In[53]:=
```
uu2[0, 0]
```
Out[53]= $\frac{\pi}{2}$

Test of the approximation:

In[54]:=
```
Plot[{ff2[θ], uu2[rR, θ]}, {θ, -π, π},
 PlotStyle → {{Thickness[0.007], Blue}, {Thickness[0.004], Red}}]
```
Out[54]=



Quite good, visually! But let us check the absolute value of the difference

In[55]:=
```
Plot[Abs[ff2[θ] - uu2[rR, θ]], {θ, -π, π},
 PlotStyle → {{Thickness[0.004], Red}}, PlotRange → All]
```
Out[55]=



I experimented and found out that the following view point works well for the plot below.

In[56]:=
```
vp2 = {2.9758612046736133`, 0.9582059152439455`, 1.2947735379246905`}
```
Out[56]= {2.97586, 0.958206, 1.29477}

In[57]:=
```
pluu2 = ParametricPlot3D[ {r Cos[θ], r Sin[θ], uu2[r, θ]},
      {r, 0, rR}, {θ, -π, π}, Mesh → False,
   PlotRange → {{-1, 1}, {-1, 1}, {0, Pi}}, BoxRatios → {1, 1, 1},
   PlotPoints → {20, 50}, ImageSize → 300, ViewPoint → Dynamic[vp2]]
```

Out[57]=



I want to show the unit circle on the graph as well. So, I define it separately. Below is the unit circle placed at the level below the minimum temperature.

In[58]:=
```
uc2 = ParametricPlot3D[{rR Cos[θ], rR Sin[θ], -.3},
   {θ, -Pi, 2 Pi}, PlotStyle → Thickness[0.005]]
```

Out[58]=



In[59]:=
```
ic2 = ParametricPlot3D[{rR Cos[θ], rR Sin[θ], ff2[θ]},
  {θ, -Pi, Pi}, PlotStyle → {Thickness[0.01], RGBColor[0, 1, 0]},
  PlotPoints → 50, BoxRatios → {1, 1, 1}, ViewPoint → vp2]
```

Out[59]=



In[60]:=
```
Show[pluu2, ic2, uc2, BoxRatios → {1, 1, 1},
 ViewPoint → vp2, PlotRange → {{-1, 1}, {-1, 1}, {-.4, 3.2}}]
```

Out[60]=



In[61]:=
```
arctan[0., 0.] = 0;
arctan[x_, y_] := ArcTan[x, y];
DensityPlot[uu2[Sqrt[x^2 + y^2], arctan[x, y]],
    {x, -1, 1}, {y, -1, 1}, PlotRange → {0, Pi}, PlotPoints → {20, 20},
 RegionFunction → Function[{x, y, z}, Norm[{x, y}] < 1.], Axes → False,
 Frame → True, LightingAngle → {0, Pi / 6}, ImageSize → 300]
```

Out[61]=



In[62]:=
```
DensityPlot[uu2[Sqrt[x^2 + y^2], arctan[x, y]],
    {x, -1, 1}, {y, -1, 1}, PlotRange → {0, Pi}, PlotPoints → {30, 30},
 RegionFunction → Function[{x, y, z}, Norm[{x, y}] < 1.], Axes → False,
 Frame → True, ColorFunction → (RGBColor[1, 1 - #^2, 1 - #^2] &),
 LightingAngle → {0, Pi / 3}, Epilog → {Circle[]}, ImageSize → 300]
```

Out[62]=



Mathematica implementation of the solution:

Example 3

Here is our function $f(\theta)$ which we will approximate with the first 15 (or nn) terms in its Fourier Series.

In[63]:=
```
Clear[ff3];
ff3[θ_] = θ^2 (Pi^2 - θ^2)
```

Out[64]= $\theta^2 \left(\pi^2 - \theta^2\right)$

Test that the function is continuous on the unit circle and that it has continuous derivative.

In[65]:=
```
test = FullSimplify[{(ff3[θ] /. {θ → -π}), (ff3[θ] /. {θ → π}),
   (D[ff3[θ], θ] /. {θ → -π}), (D[ff3[θ], θ] /. {θ → π})}]
```

Out[65]= $\{0, 0, 2\pi^3, -2\pi^3\}$

So, the derivative is not continuous, but that should not be a problem

In[66]:=
```
Plot[{ff3[θ]}, {θ, -π, π}, Ticks → {Range[-Pi, Pi, Pi / 4], Automatic}]
```

Out[66]=



Since the function that we have chosen is simple we will calculate the Fourier coefficients exactlylly.

In[67]:=
```
FullSimplify[1/π * Integrate[Expand[ff3[θ] * Cos[n θ]], {θ, -π, π}], And[n ∈ Integers]]
```

Out[67]= $-\frac{4 (-1)^n \left(-12 + n^2 \pi^2\right)}{n^4}$

In[68]:=
```
Table[-(4 (-1)^n (-12 + n^2 π^2))/n^4, {n, 1, 20}]
```

Out[68]=
$$\left\{4\left(-12+\pi^2\right),\ \frac{1}{4}\left(12-4\pi^2\right),\ \frac{4}{81}\left(-12+9\pi^2\right),\ \frac{1}{64}\left(12-16\pi^2\right),\ \frac{4}{625}\left(-12+25\pi^2\right),\ \frac{1}{324}\left(12-36\pi^2\right),\ \frac{4\left(-12+49\pi^2\right)}{2401},\ \frac{12-64\pi^2}{1024},\ \frac{4\left(-12+81\pi^2\right)}{6561},\ \frac{12-100\pi^2}{2500},\ \frac{4\left(-12+121\pi^2\right)}{14\,641},\ \frac{12-144\pi^2}{5184},\ \frac{4\left(-12+169\pi^2\right)}{28\,561},\ \frac{12-196\pi^2}{9604},\ \frac{4\left(-12+225\pi^2\right)}{50\,625},\ \frac{12-256\pi^2}{16\,384},\ \frac{4\left(-12+289\pi^2\right)}{83\,521},\ \frac{12-324\pi^2}{26\,244},\ \frac{4\left(-12+361\pi^2\right)}{130\,321},\ \frac{12-400\pi^2}{40\,000}\right\}$$

In[69]:=
```
FullSimplify[1/π * Integrate[Expand[ff3[θ] * Sin[n θ]], {θ, -π, π}], And[n ∈ Integers]]
```

Out[69]= 0

This is the approximation for the solution uu2 with 20 terms.

In[70]:=
```
rR3 = 1; nn3 = 60; Clear[uu3];

uu3[r_, θ_] = 1/(2 π) Integrate[ff3[θ], {θ, -π, π}] +
    Sum[-(4 (-1)^n (-12 + n^2 π^2))/n^4 * r^n/rR3^n * Cos[(n) θ], {n, 1, nn3}];
```

In[72]:=
```
uu3[1 / 2, π / 2]
```

Out[72]=
$$\frac{2\pi^4}{15}+\frac{12-3600\pi^2}{3\,735\,465\,674\,926\,184\,202\,240\,000}+\frac{12-3136\pi^2}{177\,162\,530\,083\,906\,533\,720\,064}+\frac{12-2704\pi^2}{8\,232\,147\,773\,269\,039\,120\,384}+\frac{12-2304\pi^2}{373\,546\,567\,492\,618\,420\,224}+\frac{12-1936\pi^2}{16\,484\,300\,536\,082\,857\,984}+\frac{12-1600\pi^2}{703\,687\,441\,776\,640\,000}+\frac{12-1296\pi^2}{28\,855\,583\,159\,353\,344}+\frac{12-1024\pi^2}{1\,125\,899\,906\,842\,624}+\frac{12-784\pi^2}{41\,248\,865\,910\,784}+\frac{12-576\pi^2}{1\,391\,569\,403\,904}+\frac{12-400\pi^2}{41\,943\,040\,000}+\frac{12-256\pi^2}{1\,073\,741\,824}+\frac{12-144\pi^2}{21\,233\,664}+\frac{12-64\pi^2}{262\,144}+\frac{12-16\pi^2}{1024}+\frac{1}{16}\left(-12+4\pi^2\right)+\frac{-12+36\pi^2}{20\,736}+\frac{-12+100\pi^2}{2\,560\,000}+\frac{-12+196\pi^2}{157\,351\,936}+\frac{-12+324\pi^2}{6\,879\,707\,136}+\frac{-12+484\pi^2}{245\,635\,219\,456}+\frac{-12+676\pi^2}{7\,666\,785\,058\,816}+\frac{-12+900\pi^2}{217\,432\,719\,360\,000}+\frac{-12+1156\pi^2}{5\,739\,519\,416\,467\,456}+\frac{-12+1444\pi^2}{143\,289\,454\,843\,396\,096}+\frac{-12+1764\pi^2}{3\,421\,345\,934\,104\,068\,096}+\frac{-12+2116\pi^2}{78\,768\,238\,957\,686\,685\,696}+\frac{-12+2500\pi^2}{1\,759\,218\,604\,441\,600\,000\,000}+\frac{-12+2916\pi^2}{38\,294\,359\,833\,110\,460\,235\,776}+\frac{-12+3364\pi^2}{815\,439\,474\,699\,835\,336\,032\,256}$$

In[73]:=
```
uu3[0, 0]
```

Out[73]=
$$\frac{2\pi^4}{15}$$

Test of the approximation:

In[74]:=
```
Plot[{ff3[θ], Evaluate[uu3[rR3, θ]]}, {θ, -π, π},
 PlotStyle → {{Thickness[0.007], Blue}, {Thickness[0.004], Red}}, PlotRange → All]
```

Out[74]=



OK, visually! But let us check the absolute value of the difference

In[75]:=
```
Plot[Abs[ff3[θ] - uu3[rR3, θ]], {θ, -π, π},
 PlotStyle → {{Thickness[0.004], Red}}, PlotRange → All]
```

Out[75]=



Not so good!

I experimented and found out that the following view point works well for the plot below.

In[76]:=
```
vp3 = {2.0478688112617216`, 2.5012750616365884`, 1.0001016937773777`}
```

Out[76]=
```
{2.04787, 2.50128, 1.0001}
```

In[77]:=
```
pluu3 = ParametricPlot3D[ {r Cos[θ], r Sin[θ], uu3[r, θ]},
      {r, 0, rR3}, {θ, -π, π}, Mesh → False,
  PlotRange → {{-1, 1}, {-1, 1}, {0, 25}}, BoxRatios → {1, 1, 1},
  PlotPoints → {20, 50}, ImageSize → 300, ViewPoint → Dynamic[vp3]]
```

Out[77]=



I want to show the unit circle on the graph as well. So, I define it separately. Below is the unit circle placed at the level below the minimum temperature.

In[78]:=
```
uc3 = ParametricPlot3D[{rR Cos[θ], rR Sin[θ], -.3},
  {θ, -Pi, 2 Pi}, PlotStyle → Thickness[0.005]]
```

Out[78]=



In[79]:=
```
ic3 = ParametricPlot3D[{rR Cos[θ], rR Sin[θ], ff3[θ]},
  {θ, -Pi, Pi}, PlotStyle → {Thickness[0.01], RGBColor[0, 1, 0]},
  PlotPoints → 50, BoxRatios → {1, 1, 1}, ViewPoint → vp2]
```

Out[79]=



In[80]:=
```
Show[pluu3, ic3, uc3, BoxRatios → {1, 1, 1},
 ViewPoint → vp3, PlotRange → {{-1, 1}, {-1, 1}, {-.4, 25}}]
```

Out[80]=



In[81]:=
```
arctan[0., 0.] = 0;
arctan[x_, y_] := ArcTan[x, y];
DensityPlot[uu3[Sqrt[x^2 + y^2], arctan[x, y]],
    {x, -1, 1}, {y, -1, 1}, PlotRange → {-.6, 25}, PlotPoints → {20, 20},
 RegionFunction → Function[{x, y, z}, Norm[{x, y}] < 1.], Axes → False,
 Frame → True, LightingAngle → {0, Pi / 6}, ImageSize → 300]
```

Out[81]=



In[82]:=
```
DensityPlot[uu3[Sqrt[x^2 + y^2], arctan[x, y]],
    {x, -1, 1}, {y, -1, 1}, PlotRange → {-.2, 25}, PlotPoints → {30, 30},
 RegionFunction → Function[{x, y, z}, Norm[{x, y}] < 1.], Axes → False,
 Frame → True, ColorFunction → (RGBColor[1, 1 - #^2, 1 - #^2] &),
 LightingAngle → {0, Pi / 3}, Epilog → {Circle[]}, ImageSize → 300]
```

Out[82]=